
F08 – Least Squares and Eigenvalue Problems (LAPACK)

F08FSF (CHETRD/ZHETRD) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F08FSF (CHETRD/ZHETRD) reduces a complex Hermitian matrix to tridiagonal form.

2 Specification

SUBROUTINE F08FSF(UPLO, N, A, LDA, D, E, TAU, WORK, LWORK, INFO)
ENTRY chetrd(UPLO, N, A, LDA, D, E, TAU, WORK, LWORK, INFO)
INTEGER N, LDA, LWORK, INFO
real D(∗), E(∗)
complex A(LDA,∗), TAU(∗), WORK(LWORK)
CHARACTER∗1 UPLO

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine reduces a complex Hermitian matrix A to real symmetric tridiagonal form T by a unitary
similarity transformation: A = QTQH.

The matrix Q is not formed explicitly but is represented as a product of n− 1 elementary reflectors (see
the Chapter Introduction for details). Routines are provided to work with Q in this representation (see
Section 8).

4 References

[1] Golub G H and van Loan C F (1996) Matrix Computations Johns Hopkins University Press (3rd
Edition), Baltimore

5 Parameters

1: UPLO — CHARACTER*1 Input

On entry: indicates whether the upper or lower triangular part of A is stored as follows:

if UPLO = ’U’, then the upper triangular part of A is stored;
if UPLO = ’L’, then the lower triangular part of A is stored.

Constraint: UPLO = ’U’ or ’L’.

2: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: A(LDA,∗) — complex array Input/Output

Note: the second dimension of the array A must be at least max(1,N).

On entry: the n by n Hermitian matrix A. If UPLO = ’U’, the upper triangle of A must be stored
and the elements of the array below the diagonal are not referenced; if UPLO = ’L’, the lower
triangle of A must be stored and the elements of the array above the diagonal are not referenced.

On exit: A is overwritten by the tridiagonal matrix T and details of the unitary matrix Q as specified
by UPLO.

4: LDA — INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08FSF
(CHETRD/ZHETRD) is called.

Constraint: LDA ≥ max(1,N).

5: D(∗) — real array Output

Note: the dimension of the array D must be at least max(1,N).

On exit: the diagonal elements of the tridiagonal matrix T .

6: E(∗) — real array Output

Note: the dimension of the array E must be at least max(1,N−1).

On exit: the off-diagonal elements of the tridiagonal matrix T .

7: TAU(∗) — complex array Output

Note: the dimension of the array TAU must be at least max(1,N−1).

On exit: further details of the unitary matrix Q.

8: WORK(LWORK) — complex array Workspace

On exit: if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum
performance.

9: LWORK — INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F08FSF
(CHETRD/ZHETRD) is called.

Suggested value: for optimum performance LWORK should be at least N × nb, where nb is the
blocksize.

Constraint: LWORK ≥ 1.

10: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix A + E, where

‖ E ‖2 ≤ c(n)ε‖ A ‖2,

c(n) is a modestly increasing function of n, and ε is the machine precision.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the
computation, but this does not affect the stability of the eigenvalues and eigenvectors.

8 Further Comments

The total number of real floating-point operations is approximately 16n3

3 .

To form the unitary matrix Q this routine may be followed by a call to F08FTF (CUNGTR/ZUNGTR):

CALL CUNGTR (UPLO,N,A,LDA,TAU,WORK,LWORK,INFO)

To apply Q to an n by p complex matrix C this routine may be followed by a call to F08FUF
(CUNMTR/ZUNMTR). For example,

CALL CUNMTR (’Left’,UPLO,’No Transpose’,N,P,A,LDA,TAU,C,LDC,
+ WORK,LWORK,INFO)

forms the matrix product QC.

The real analogue of this routine is F08FEF (SSYTRD/DSYTRD).

9 Example

To reduce the matrix A to tridiagonal form, where

A =



−2.28 + 0.00i 1.78 − 2.03i 2.26 + 0.10i −0.12 + 2.53i

1.78 + 2.03i −1.12 + 0.00i 0.01 + 0.43i −1.07 + 0.86i
2.26 − 0.10i 0.01 − 0.43i −0.37 + 0.00i 2.31 − 0.92i

−0.12 − 2.53i −1.07 − 0.86i 2.31 + 0.92i −0.73 + 0.00i


 .

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F08FSF Example Program Text
* Mark 16 Release. NAG Copyright 1992.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, LDA, LWORK
PARAMETER (NMAX=8,LDA=NMAX,LWORK=64*NMAX)

* .. Local Scalars ..
INTEGER I, INFO, J, N
CHARACTER UPLO

* .. Local Arrays ..
complex A(LDA,NMAX), TAU(NMAX-1), WORK(LWORK)
real D(NMAX), E(NMAX-1)

* .. External Subroutines ..
EXTERNAL chetrd

* .. Executable Statements ..
WRITE (NOUT,*) ’F08FSF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read A from data file
*

READ (NIN,*) UPLO
IF (UPLO.EQ.’U’) THEN

READ (NIN,*) ((A(I,J),J=I,N),I=1,N)

ELSE IF (UPLO.EQ.’L’) THEN
READ (NIN,*) ((A(I,J),J=1,I),I=1,N)

END IF
*
* Reduce A to tridiagonal form
*

CALL chetrd(UPLO,N,A,LDA,D,E,TAU,WORK,LWORK,INFO)
*
* Print tridiagonal form
*

WRITE (NOUT,*)
WRITE (NOUT,*) ’Diagonal’
WRITE (NOUT,99999) (D(I),I=1,N)
WRITE (NOUT,*) ’Off-diagonal’
WRITE (NOUT,99999) (E(I),I=1,N-1)

END IF
STOP

*
99999 FORMAT (1X,8F9.4)

END

9.2 Program Data

F08FSF Example Program Data
4 :Value of N
’L’ :Value of UPLO

(-2.28, 0.00)
( 1.78, 2.03) (-1.12, 0.00)
( 2.26,-0.10) ( 0.01,-0.43) (-0.37, 0.00)
(-0.12,-2.53) (-1.07,-0.86) ( 2.31, 0.92) (-0.73, 0.00) :End of matrix A

9.3 Program Results

F08FSF Example Program Results

Diagonal
-2.2800 -0.1285 -0.1666 -1.9249

Off-diagonal
-4.3385 -2.0226 -1.8023

F08FSF (CHETRD/ZHETRD).4 (last) [NP3390/19/pdf]


